
The goal of the PBM is to provide the absolute best care in terms of medication for the members while reducing the cost of medication expense borne by the Health Plan and the members.  In this way both patient outcomes and Health Care affordability can be maximized and as a result, society benefits.
Typically Physician Profiling Reports are utilized in medical claims benefit management to compare physician services with other peer physicians in similar specialties to monitor various metrics and provide incentives to physicians for cost effective strategies which help lower Benefit Plan costs.  Performance Measures are focused more on medical procedures and visits, rather than medication prescribing.  However, Prescriber Profiling focused on prescription drugs is also very useful.
In order for Clinical Pharmacists to use the data efficiently to improve patient outcomes and reduce cost, the data must be transformed into actionable information.  This paper outlines very specifically utilizing the power of SAS programming Data Step and PROC Step methodology to convert large quantities of data into an organized and meaningful Prescriber Profiling report focused on medication prescribing that can be used to achieve these goals. 
Analysis of prescriber metrics compared to the entire Plan metrics is important to this process.  Therefore I selected the following eight industry standard metrics to study, at the prescriber level as compared to the average Plan level:
1. Generic Utilization Rate
a. generic drugs divided by all brand and generic drugs
b. low generic utilization rates are problematic because brand drugs cost more and are not always more effective
2. Formulary Compliance 
a. drugs on plan formulary divided by formulary plus non-formulary drugs
b. formulary drugs are deemed more effective and in some cases less costly
3. Brand Cost Per Rx
a. brand drug cost divided by brand drug prescription count
b. higher than average brand drug cost per prescription should be researched 
4. Generic Cost Per Rx 
a. generic drug cost divided by generic drug prescription count
b. higher than average generic drug cost per prescription should be researched

5. Specialty Cost Per Rx 
a. specialty drug cost divided by specialty drug prescription count
b. higher than average specialty drug cost per prescription should be researched
6. Hep C Drug Utilization
a. percent of Hep C drug claims
b. higher than average Hep C drug claim percentage should be researched
7. Specialty Drug Utilization
a. percent of higher cost Specialty Drug claims
b. higher than average Specialty Drug claim percentage should be researched
8. Opioid Drug Utilization 
a. percent of Opioid drug claims
b. higher than average Opioid drug claim percentage should be researched
In the recent past, although non-Specialty drug cost trends are somewhat flat or slightly increasing from year to year, higher cost Specialty drug cost trends, especially Hep C treatment drugs have risen materially and it is important to manage these costs while providing the best possible patient care.  Given the recent and dangerous epidemic of Opioid overuse, it is imperative to track Opioid use from a member safety standpoint and intervene where necessary.  
In order to leverage our resource assets to the maximize advantage, it is important to reduce the vast number of prescriber data reviewed to a manageable amount.   Researching every prescriber metric determined to be above average for every Plan would be a momentous and unreasonable exercise. Therefore I came up with an idea, specifically to produce one warning flag per prescriber for each of the above eight metrics where the prescriber is an outlier, accumulate the warning count 1 to 8, and create Tiers of prescribers to review based on this warning count.  For example, if one prescriber is an outlier in 8 of the above 8 metrics, they are assigned to Tier 8 and are considered the top risk priority.  Prescribers that are an outlier in 7 of 8 metrics go to Tier 7, and so on and so forth.  Tier 0, the largest group by far, are prescribers with outliers in 0 out of 8 metrics and are zero risk for this process.
Another important metric is percent spend per prescriber, in other words what percent cost to total spend is attributed to each individual prescriber.  For example if there are 10 prescribers in the Tier 8 category, it makes sense to sort in descending percent spend order and review prescribers who impact the total dollar spend the most.  If Dr. Smith has 2% of total spend and Dr. Jones has .0045% of total spend, my resources would be better spent to review Dr. Smith’s data rather than Dr. Jones’ data.
The first step in creating this logic is to pull the data from the company Data Warehouse.  In order to improve flexibility for running various customers and various date spans, macro variables for customer number, beginning fill date and end fill date are created.  
LIBNAME odb ODBC DSN=InformDB user=InformaticsUser pw=P@ssw0rd schema=dbo;
LIBNAME PRESCR "I:\SAS\Projects";

*** CREATE MACRO VARIABLES ***;

%let Customer  = 603;
%let StartDate = '20180901';
%let EndDate   = '20180930';

  
*** READ DATA FROM DATA WAREHOUSE TO CREATE PRESCRIBER PROFILING REPORT **************************;

DATA PRESCR.PLAN(KEEP = CUSTOMERID CLAIMNO GCN NDC FILLDT PAIDAMT CICSCLAIMSSTATUSCODE CLAIMTYPE
                 PLANMETQTY PLANDAYSSUPPLY PRESCRIBERID PRESCRIBERLASTNAME DRUGCATEGORYCODE
                 DRUGNAME SPECIFICTHERAPEUTICCD GI GPI FORMULARYIND MAINTENANCEDRUGIND COMPOUNDCODE
                 PRESCRIBERIDNPI DME PRESCRIBERSPECIALTY BRANDGENERICCODE GROUPID);
   SET ODB.FinalizedFactRxClaims(WHERE=(CUSTOMERID = &Customer. AND &StartDate. <= FILLDT <= &EndDate. ));
   
RUN;

PROC SORT DATA = PRESCR.PLAN;
   BY GCN;
RUN;

The next step is to identify specialty drugs paid with the specialty benefit, and update each data record by merging by the drug indicator Generic Code Number (GCN).  Note we create a SPECIALTY flag and value with “Y” or “N”.
*** IDENTIFY DRUGS ADJUDICATED WITH THE SPECIALTY RATE USING PERFORMRX SPECIALTY LIST ***;

PROC IMPORT OUT = WORK.SPECIALTY
      DATAFILE = "T:\Pharmacy_Informatics\Informatics\Reference\Specialty and Injectables\
                  Specialty Formulary - Perform Specialty_2018-11-06.xlsx"
      DBMS=EXCELCS REPLACE;
      SHEET = "GCN_LIST";
RUN;

DATA PRESCR.SPEC (KEEP = GCN);
   SET SPECIALTY;
   IF GCN = ' ' THEN DELETE;
RUN;

PROC SORT DATA = PRESCR.SPEC;
   BY GCN;
RUN;

DATA PLAN2 SPECONLY;
   MERGE PRESCR.PLAN (IN=A) PRESCR.SPEC (IN=B);
   BY GCN;
   IF A AND B THEN
   DO;
      SPECIALTY = "Y";
      OUTPUT PLAN2;
   END;
   ELSE IF A AND NOT B THEN
   DO;
      SPECIALTY = "N";
      OUTPUT PLAN2;
   END;
   ELSE IF B AND NOT A THEN OUTPUT SPECONLY;
RUN;
  The next step is to merge to the First Data Bank drug information database by National Drug Code (NDC), another drug identifier, and add some important drug identifiers that will be useful to our Clinicians reviewing the data.  Note since all drugs in our claim data should also match the entire population of possible drugs, I have included an ABORT statement to abend the SAS Program if a non-match is encountered.  In the event of an ABORT, this would force manual research to ensure validity of results.
*** MERGE TO FIRST DATA BANK DATABASE TO ADD CRITICAL DRUG IDENTIFICATION FIELDS ***;

PROC IMPORT OUT = WORK.FDB (KEEP = NDC BN LN GNN INNOV AHFS8 AHFS_DESCRIPTION HIC3_DESC)
      DATAFILE = "T:\Pharmacy_Informatics\Informatics\FDB\FDB_MEDSP_CROSSWALK\FDB3YR_MEDSP_MSTR_20181112.xlsx"
      DBMS=EXCELCS REPLACE;
      SHEET = "FDBMED_3YRS";
RUN;

PROC SORT DATA = FDB;
   BY NDC;
RUN;

PROC SORT DATA = PLAN2;
   BY NDC;
RUN;







*** ALL CLAIM RECORDS SHOULD MATCH TO THE FIRST DATA BANK DATABASE ***;
*** ABORT PROGRAM FOR RESEARCH PURPOSES IF DRUG RECORDS ARE ENCOUNTERED THAT DO NOT MATCH TO THE FIRST DATABANK DATABASE ***;

DATA PLAN3 FDBONLY;
   MERGE PLAN2 (IN=A) FDB (IN=B);
   BY NDC;
   IF A AND B THEN OUTPUT PLAN3;
   ELSE IF A AND NOT B THEN ABORT;
   ELSE IF B AND NOT A THEN OUTPUT FDBONLY;
RUN;
Now we need to identify Opioid drugs and create an OPIOID flag with values “Y” and “N”.
*** MERGE TO OPIOID DRUG IDENTIFICATION FILE IN ORDER TO IDENTIFY OPIOIDS ***;

PROC IMPORT OUT = WORK.OPIOIDS (KEEP = GCN)
      DATAFILE = "I:\SAS\Projects\Opioids.xlsx"
      DBMS=EXCELCS REPLACE;
      SHEET = "OPIOIDS";
RUN;

PROC SORT DATA = OPIOIDS;
   BY GCN;
RUN;

PROC SORT DATA = PLAN3;
   BY GCN;
RUN;

DATA PLAN3A OPIOIDSONLY;
   MERGE PLAN3 (IN=A) OPIOIDS (IN=B);
   BY GCN;
   IF A AND B THEN
   DO;
      OPIOIDS = "Y";
      OUTPUT PLAN3A;
   END;
   ELSE IF A AND NOT B THEN
   DO;
      OPIOIDS = "N";
      OUTPUT PLAN3A;
   END;
   ELSE IF B AND NOT A THEN OUTPUT OPIOIDSONLY;
RUN;

Next we process records, create a SAS date using the MDY function, categorize drugs as brand, generic or supplies, assign Hep C drugs and create a HEPC flag valued “Y” or “N” and start building various drug counts to use as a basis for calculating metrics.
*** PROCESS RECORDS ***;
DATA PLAN4;
   SET PLAN3A;
   FORMAT DATE_FILLED MONYY5.;

   BETA=1;

  *** CREATE SAS DATE ***;

   MM = SUBSTR(FILLDT,6,2);
   DD = SUBSTR(FILLDT,9,2);
   YY = SUBSTR(FILLDT,1,4);
   DATE_FILLED = MDY(MM,DD,YY);

   RXCOUNT = 1;

   *** DETERMINE BRAND VERSUS GENERIC DRUGS ***;

   IF GPI = "0"                           THEN DRUG_TYPE = "DME SUPPLIES";
   ELSE IF GI = "2" AND GPI = "2"         THEN DRUG_TYPE = "BRAND";
   ELSE IF GI = '1' AND GPI = '1'         THEN DRUG_TYPE = "GENERIC";
   ELSE IF GI = '1' AND GPI = '2'         THEN DRUG_TYPE = "BRAND";
   ELSE IF GI = '2' AND GPI = '1'         THEN DRUG_TYPE = "GENERIC";

   *** ASSIGN HEP C DRUGS ***;

   HEPC = 'N';
   IF GCN IN ('29964' '29941' '35708' '35648' '37179' '37614' '37073' 
              '37074' '37844' '40615' '41729' '41335' '41932' '43634' '43699') THEN HEPC = "Y";

   *** SET COUNTS AND AMOUNTS TO ZERO ***;

   FORMULARY_COUNT = 0;
   NONFORMULARY_COUNT = 0;
   BRAND_COUNT = 0;
   BRAND_AMT = 0;
   GENERIC_COUNT = 0;
   GENERIC_AMT = 0;
   HEPC_COUNT = 0;
   HEPC_AMT = 0;
   SPECIALTY_COUNT = 0;
   SPECIALTY_AMT = 0;
   OPIOID_COUNT = 0;
   OPIOID_AMT = 0;

   IF      FORMULARYIND = 1 THEN FORMULARY_COUNT    = RXCOUNT;
   ELSE IF FORMULARYIND = 2 THEN NONFORMULARY_COUNT = RXCOUNT;
   
   *** BRAND DRUGS NOT SPECIALTY ***;

   IF DRUG_TYPE = "BRAND" AND SPECIALTY = "N" THEN
   DO;
      BRAND_COUNT = RXCOUNT;
      BRAND_AMT   = PAIDAMT;
   END;

   *** GENERIC DRUGS NOT SPECIALTY ***;

   ELSE IF DRUG_TYPE = "GENERIC" AND SPECIALTY = "N" THEN
   DO;
      GENERIC_COUNT = RXCOUNT;
      GENERIC_AMT   = PAIDAMT;
   END;

   *** HEP C DRUGS ***;

   IF HEPC = "Y" THEN
   DO;
      HEPC_COUNT = RXCOUNT;
      HEPC_AMT   = PAIDAMT;
   END;

   *** SPECIALTY DRUGS ***;

   IF SPECIALTY = "Y" THEN
   DO;
      SPECIALTY_COUNT = RXCOUNT;
      SPECIALTY_AMT   = PAIDAMT;
   END;

   *** OPIOID DRUGS ***;

   IF OPIOIDS ="Y" THEN
   DO;
      OPIOID_COUNT = RXCOUNT;
      OPIOID_AMT   = PAIDAMT;
   END;
   
RUN;



We now need to summarize by prescriber identification number using a PROC SUMMARY procedure.
*** SUMMARIZE AT PRESCRIBER LEVEL ***;

PROC SUMMARY NWAY MISSING DATA = PLAN4;
   CLASS PRESCRIBERID;
   VAR RXCOUNT PAIDAMT FORMULARY_COUNT NONFORMULARY_COUNT BRAND_COUNT BRAND_AMT
       GENERIC_COUNT GENERIC_AMT HEPC_COUNT HEPC_AMT SPECIALTY_COUNT SPECIALTY_AMT
       OPIOID_COUNT OPIOID_AMT;
   ID PRESCRIBERLASTNAME BETA;
   OUTPUT OUT = PLAN4_SUMMM (DROP = _TYPE_ _FREQ_)SUM=;
RUN;

Next we calculate all prescriber metrics for this project which are Generic Utilization Rate, Formulary Compliance, Percent Hep C, Percent Specialty, Percent Opioid, Brand Cost Per Rx, Generic Cost Per Rx, and Specialty Cost Per Rx.
*** CALCULATE ALL PRESCRIBER METRICS ***;

DATA PLAN5;
   SET PLAN4_SUMMM;
   
   IF (BRAND_COUNT + GENERIC_COUNT) NE 0 THEN GUR = (GENERIC_COUNT/(BRAND_COUNT + GENERIC_COUNT));
   ELSE IF (BRAND_COUNT + GENERIC_COUNT) = 0 THEN GUR = 0;
   IF (FORMULARY_COUNT + NONFORMULARY_COUNT) NE 0 
      THEN FORMULARY_COMPLIANCE = (FORMULARY_COUNT / (FORMULARY_COUNT + NONFORMULARY_COUNT));
   ELSE IF (FORMULARY_COUNT + NONFORMULARY_COUNT) = 0 THEN FORMULARY_COMPLIANCE = 0;
   PERCENT_HEPC = (HEPC_COUNT / RXCOUNT);
   PERCENT_SPEC = (SPECIALTY_COUNT / RXCOUNT);
   IF BRAND_COUNT NE 0 THEN BRANDCOST_RX = (BRAND_AMT/BRAND_COUNT);
   ELSE IF BRAND_COUNT = 0 THEN BRANDCOST_RX = 0;
   IF GENERIC_COUNT NE 0 THEN GENERICCOST_RX = (GENERIC_AMT/GENERIC_COUNT);
   ELSE IF GENERIC_COUNT = 0 THEN GENERICCOST_RX = 0;
   IF SPECIALTY_COUNT NE 0 THEN SPECIALTYCOST_RX = (SPECIALTY_AMT/SPECIALTY_COUNT);
   ELSE IF SPECIALTY_COUNT = 0 THEN SPECIALTYCOST_RX = 0;
   PERCENT_OPIOID = (OPIOID_COUNT / RXCOUNT);
RUN;




We now need to summarize at the Plan level, so we can compare Prescriber versus Plan per metric and identify outliers.
*** SUMMARIZE AT PLAN LEVEL TO MAKE COMPARISONS ***;

PROC SUMMARY NWAY MISSING DATA = PLAN4;
   CLASS BETA;
   VAR RXCOUNT PAIDAMT FORMULARY_COUNT NONFORMULARY_COUNT BRAND_COUNT
       BRAND_AMT GENERIC_COUNT GENERIC_AMT HEPC_COUNT HEPC_AMT SPECIALTY_COUNT
       SPECIALTY_AMT OPIOID_COUNT OPIOID_AMT;
   OUTPUT OUT = PLANTOT4_SUMMM (DROP = _TYPE_ _FREQ_)SUM=;
RUN;
Next we calculate the metrics at the Plan level.
*** CALCULATE ALL PLAN METRICS ***;

DATA PLANTOT5 (KEEP = BETA PLANRXCOUNT PLANPAIDAMT PLANFORMULARY_COUNT PLANNONFORMULARY_COUNT
                      PLANBRAND_COUNT PLANBRAND_AMT PLANGENERIC_COUNT PLANGENERIC_AMT PLANHEPC_COUNT
                      PLANHEPC_AMT PLANSPECIALTY_COUNT PLANSPECIALTY_AMT PLANOPIOID_COUNT
                      PLANOPIOID_AMT PLANGUR PLANFORMULARY_COMPLIANCE PLANPERCENT_HEPC PLANPERCENT_SPEC
                      PLANBRANDCOST_RX PLANGENERICCOST_RX PLANSPECIALTYCOST_RX PLANPERCENT_OPIOID);
   SET PLANTOT4_SUMMM;
   PLANRXCOUNT = RXCOUNT;
   PLANPAIDAMT = PAIDAMT;
   PLANFORMULARY_COUNT = FORMULARY_COUNT;
   PLANNONFORMULARY_COUNT = NONFORMULARY_COUNT;
   PLANBRAND_COUNT = BRAND_COUNT;
   PLANBRAND_AMT = BRAND_AMT;
   PLANGENERIC_COUNT = GENERIC_COUNT;
   PLANGENERIC_AMT = GENERIC_AMT;
   PLANHEPC_COUNT = HEPC_COUNT;
   PLANHEPC_AMT = HEPC_AMT;
   PLANSPECIALTY_COUNT = SPECIALTY_COUNT;
   PLANSPECIALTY_AMT=SPECIALTY_AMT;
   PLANGUR = (GENERIC_COUNT/(BRAND_COUNT + GENERIC_COUNT));
   PLANFORMULARY_COMPLIANCE = (FORMULARY_COUNT / (FORMULARY_COUNT + NONFORMULARY_COUNT));
   PLANPERCENT_HEPC = (HEPC_COUNT / RXCOUNT);
   PLANPERCENT_SPEC = (SPECIALTY_COUNT / RXCOUNT);
   PLANBRANDCOST_RX = (BRAND_AMT/BRAND_COUNT);
   PLANGENERICCOST_RX = (GENERIC_AMT/GENERIC_COUNT);
   PLANSPECIALTYCOST_RX = (SPECIALTY_AMT/SPECIALTY_COUNT);
   PLANOPIOID_COUNT = OPIOID_COUNT;
   PLANOPIOID_AMT = OPIOID_AMT;
   PLANPERCENT_OPIOID = (OPIOID_COUNT / RXCOUNT);
RUN;
Next we merge the prescriber level information and the plan level information together, using a field created called BETA, so each prescriber record will also have the total Plan metrics for comparison purposes.  Additionally we create our PercentSpend field, we create our 8 Warning fields, we compare prescriber to Plan at the metric level, we increment our FLAG value based on the number of warnings (1-8) and we create our review Tiers.   Note again we ABORT the program if there is no match.
*** ALL RECORDS SHOULD MERGE OR ABEND AND RESEARCH ERROR ***;

DATA PLAN6;
   FORMAT PRESCRIBERID PRESCRIBERLASTNAME $15. RXCOUNT PLANRXCOUNT COMMA8. 
          PAIDAMT PLANPAIDAMT DOLLAR15.2 FORMULARY_COUNT PLANFORMULARY_COUNT 
          NONFORMULARY_COUNT PLANNONFORMULARY_COUNT COMMA8.
          FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE PERCENT9.2 
          BRAND_COUNT PLANBRAND_COUNT GENERIC_COUNT PLANGENERIC_COUNT COMMA8.
          GUR PLANGUR PERCENT9.2 BRAND_AMT PLANBRAND_AMT 
          BRANDCOST_RX PLANBRANDCOST_RX GENERIC_AMT PLANGENERIC_AMT GENERICCOST_RX
          PLANGENERICCOST_RX SPECIALTY_AMT PLANSPECIALTY_AMT SPECIALTYCOST_RX 
          PLANSPECIALTYCOST_RX DOLLAR15.2 SPECIALTY_COUNT PLANSPECIALTY_COUNT COMMA8. 
          PERCENT_SPEC PLANPERCENT_SPEC PERCENT9.2 HEPC_COUNT PLANHEPC_COUNT COMMA8. 
          PERCENT_HEPC PLANPERCENT_HEPC PERCENT9.2 HEPC_AMT PLANHEPC_AMT DOLLAR15.2 
          OPIOID_COUNT PLANOPIOID_COUNT COMMA8. OPIOID_AMT PLANOPIOID_AMT DOLLAR15.2 
          PERCENT_OPIOID PLANPERCENT_OPIOID PERCENTSPEND PERCENT9.2 
          WARNING1 WARNING2 WARNING3 WARNING4 WARNING5 WARNING6 WARNING7 WARNING8 $60. BETA 1.;
   MERGE PLAN5 (IN=A) PLANTOT5 (IN=B);
   BY BETA;
   IF A AND B THEN
   DO;
      PERCENTSPEND = (PAIDAMT/PLANPAIDAMT);
	  
	  *** DEVELOP WARNINGS FOR OUTLIER PRESCRIBER UTILIZATION/COST ***;

	  WARNING1 = ' ';
	  WARNING2 = ' ';
	  WARNING3 = ' ';
	  WARNING4 = ' ';
	  WARNING5 = ' ';
	  WARNING6 = ' ';
	  WARNING7 = ' ';
	  WARNING8 = ' ';

	  

  *** SET FLAG MECHANISM TO ZERO ***;

	  FLAG = 0;

	  *** INCREMENT FLAG BASED ON NUMBER OF WARNINGS ***;

	  IF GUR < PLANGUR THEN
      DO;
         FLAG + 1;
         WARNING1 = "Prescriber Generic Utilization Rate Lower Than Plan Average";
	  END;
	  IF FORMULARY_COMPLIANCE < PLANFORMULARY_COMPLIANCE THEN
      DO;
         FLAG + 1;
         WARNING2 = "Prescriber Formulary Compliance Lower Than Plan Average";
	  END;
	  IF BRANDCOST_RX > PLANBRANDCOST_RX THEN
      DO;
         FLAG + 1;
         WARNING3 = "Prescriber Brand Cost Per Rx Higher Than Plan Average";
      END;
	  IF GENERICCOST_RX > PLANGENERICCOST_RX THEN
      DO;
         FLAG + 1;
         WARNING4 = "Prescriber Generic Cost Per Rx Higher Than Plan Average";
	  END;
	  IF SPECIALTYCOST_RX > PLANSPECIALTYCOST_RX THEN
      DO;
         FLAG + 1;
         WARNING5 = "Prescriber Specialty Cost Per Rx Higher Than Plan Average";
	  END;
	  IF PERCENT_HEPC > PLANPERCENT_HEPC THEN
      DO;
         FLAG + 1;
         WARNING6 = "Prescriber Hep C Drug Utilization Higher Than Plan Average";
      END;
	  IF PERCENT_SPEC > PLANPERCENT_SPEC THEN
      DO;
         FLAG + 1;
         WARNING7 = "Prescriber Specialty Utilization Higher Than Plan Average";
	  END;
	  IF PERCENT_OPIOID > PLANPERCENT_OPIOID THEN
      DO;
        FLAG + 1;
        WARNING8 = "Prescriber Opioid Utilization Higher Than Plan Average";
	  END;

*** CREATE TIERS BASED ON NUMBER OF FLAGS FOR CLINICAL STAFF TO REVIEW TOP TIERS AND LOOK FOR ***;
*** OPPORTUNITIES TO IMPROVE PATIENT CARE/SAFETY AND DECREASE COST                            ***;
     

	  TIER = 0;
	  IF      FLAG = 8 THEN TIER = 8;
	  ELSE IF FLAG = 7 THEN TIER = 7;
	  ELSE IF FLAG = 6 THEN TIER = 6;
	  ELSE IF FLAG = 5 THEN TIER = 5;
	  ELSE IF FLAG = 4 THEN TIER = 4;
	  ELSE IF FLAG = 3 THEN TIER = 3;
	  ELSE IF FLAG = 2 THEN TIER = 2;
	  ELSE IF FLAG = 1 THEN TIER = 1;
      ELSE                  TIER = 0;

      OUTPUT PLAN6;
   END;

   *** ALL RECORDS MUST MERGE THEREFORE ABORT IF RECORDS DO NOT MERGE ***;

   ELSE IF A AND NOT B THEN ABORT;
   ELSE IF B AND NOT A THEN ABORT;
RUN;

Next we create two data sets for each Tier, one with metrics and one with all corresponding fields.  In order to create field descriptive titles for output to MS Excel, we create LABELS.  The other alternative of editing the MS Excel file tabs is very time consuming and inefficient.


*** CREATE TWO DATA SETS PER TIER, ONE STREAMLINED AND ONE WITH ALL CORRESPONDING DETAIL ***;

DATA    TIER0 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER0DETAIL
	 TIER1 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	TIER1DETAIL
       TIER2 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER2DETAIL
        TIER3 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER3DETAIL
        TIER4 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
       TIER4DETAIL
       TIER5 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER5DETAIL
	 TIER6 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER6DETAIL
	 TIER7 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER7DETAIL
	 TIER8 (KEEP = PRESCRIBERID PRESCRIBERLASTNAME PERCENTSPEND GUR PLANGUR
                   FORMULARY_COMPLIANCE PLANFORMULARY_COMPLIANCE BRANDCOST_RX
                   PLANBRANDCOST_RX GENERICCOST_RX PLANGENERICCOST_RX
                   SPECIALTYCOST_RX PLANSPECIALTYCOST_RX PERCENT_HEPC 
                   PLANPERCENT_HEPC PERCENT_SPEC PLANPERCENT_SPEC 
                   PERCENT_OPIOID PLANPERCENT_OPIOID)
	 TIER8DETAIL;

   SET PLAN6;

   *** CREATE LABELS SO OUTPUT FILE FIELD NAMING CONVENTIONS ARE DESCRIPTIVE ***;

   LABEL PRESCRIBERID             = "Prescriber ID" 
       PRESCRIBERLASTNAME       = "Prescriber Last Name"
       PERCENTSPEND             = "Prescriber Spend Compared to Total Plan"
       GUR                      = "Prescriber Generic Utilization Rate"
       PLANGUR                  = "Plan Generic Utilization Rate" 
       FORMULARY_COMPLIANCE     = "Prescriber Formulary Compliance"
       PLANFORMULARY_COMPLIANCE = "Plan Formulary Compliance"
       BRANDCOST_RX             = "Prescriber Brand Cost Per Rx"
       PLANBRANDCOST_RX         = "Plan Brand Cost Per Rx"
       GENERICCOST_RX           = "Prescriber Generic Cost Per Rx"
       PLANGENERICCOST_RX       = "Plan Generic Cost Per Rx"
	SPECIALTYCOST_RX         = "Prescriber Specialty Cost Per Rx"
       PLANSPECIALTYCOST_RX     = "Plan Specialty Cost Per Rx"
       PERCENT_HEPC             = "Prescriber Percent Hep C Drugs Prescribed"
       PLANPERCENT_HEPC         = "Plan Percent Hep C Drugs Prescribed"
       PERCENT_SPEC             = "Prescriber Percent Specialty Drugs Prescribed"
       PLANPERCENT_SPEC         = "Plan Percent Specialty Drugs Prescribed"
       PERCENT_OPIOID           = "Prescriber Percent Opioid Drugs Prescribed"
       PLANPERCENT_OPIOID       = "Plan Percent Opioid Drugs Prescribed"
	RXCOUNT                  = "Prescriber Rx Count"
	PLANRXCOUNT              = "Plan Rx Count"
       PAIDAMT                  = "Prescriber Total Paid"
       PLANPAIDAMT              = "Plan Total Paid"
	FORMULARY_COUNT          = "Prescriber Formulary Rx Count"
	PLANFORMULARY_COUNT      = "Plan Formulary Rx Count"
	NONFORMULARY_COUNT       = "Prescriber Non-Formulary Rx Count"
	PLANNONFORMULARY_COUNT   = "Plan Non-Formulary Rx Count"
	BRAND_COUNT              = "Prescriber Brand Rx Count"
	PLANBRAND_COUNT          = "Plan Brand Rx Count"
	GENERIC_COUNT            = "Prescriber Generic Rx Count"
	PLANGENERIC_COUNT        = "Plan Generic Rx Count"
	BRAND_AMT                = "Prescriber Brand Amount Paid"
	PLANBRAND_AMT            = "Plan Brand Amount Paid"
	GENERIC_AMT              = "Prescriber Generic Amount Paid"
	PLANGENERIC_AMT          = "Plan Generic Amount Paid"
	SPECIALTY_AMT            = "Prescriber Specialty Amount Paid"
	PLANSPECIALTY_AMT        = "Plan Specialty Amount Paid"
	SPECIALTY_COUNT          = "Prescriber Specialty Rx Count"
	PLANSPECIALTY_COUNT      = "Plan Specialty Rx Count"
	HEPC_COUNT               = "Prescriber Hep C Rx Count"
	PLANHEPC_COUNT           = "Plan Hep C Rx Count"
	HEPC_AMT                 = "Prescriber Hep C Amount Paid"
	PLANHEPC_AMT             = "Plan Hep C Amount Paid"
	OPIOID_COUNT             = "Prescriber Opioid Drug Rx Count"
	PLANOPIOID_COUNT         = "Plan Opioid Drug Rx Count"
	OPIOID_AMT               = "Prescriber Opioid Amount Paid"
	PLANOPIOID_AMT           = "Plan Opioid Amount Paid"
	WARNING1                 = "Warning 1"
	WARNING2                 = "Warning 2"
	WARNING3                 = "Warning 3"
	WARNING4                 = "Warning 4"
	WARNING5                 = "Warning 5"
	WARNING6                 = "Warning 6"
	WARNING7                 = "Warning 7"
	WARNING8                 = "Warning 8"
        BETA                     = "Beta"
        FLAG                     = "Flag"
        TIER                     = "Tier";


   IF      TIER = 0 THEN
   DO;
      OUTPUT TIER0;
	  OUTPUT TIER0DETAIL;
   END;
   IF      TIER = 1 THEN
   DO;
      OUTPUT TIER1;
	  OUTPUT TIER1DETAIL;
   END;
   ELSE IF TIER = 2 THEN
   DO;
      OUTPUT TIER2;
	  OUTPUT TIER2DETAIL;
   END;
   ELSE IF TIER = 3 THEN
   DO;
      OUTPUT TIER3;
	  OUTPUT TIER3DETAIL;
   END;
   ELSE IF TIER = 4 THEN
   DO;
      OUTPUT TIER4;
	  OUTPUT TIER4DETAIL;
   END;
   ELSE IF TIER = 5 THEN
   DO;
      OUTPUT TIER5;
	  OUTPUT TIER5DETAIL;
   END;
   ELSE IF TIER = 6 THEN
   DO;
      OUTPUT TIER6;
	  OUTPUT TIER6DETAIL;
   END;
   ELSE IF TIER = 7 THEN
   DO;
      OUTPUT TIER7;
	  OUTPUT TIER7DETAIL;
   END;
   ELSE IF TIER = 8 THEN
   DO;
      OUTPUT TIER8;
	  OUTPUT TIER8DETAIL;
   END;
RUN;
 
Now we need to invoke the PROC SORT procedure to sort by descending PERCENTSPEND, as the higher prescriber percent spend is more important to review.  Rather than write code to sort 16 data sets, we create an efficient macro.
*** CREATE MACRO TO SORT EACH TIER BY DESCENDING PERCENT SPEND                     ***;
*** PERCENT SPEND IS THE PERCENTAGE OF EACH PRESCRIBER'S SPEND TO TOTAL PLAN SPEND ***;
*** THIS SORT PROCEDURE ALLOWS RESOURCES TO RESEARCH THE HIGHER TIERS IN ORDER OF  ***;
*** HIGHEST DOLLAR IMPACT OF PRESCRIBER                                            ***;

%MACRO SortTiers (File= ) ;
   PROC SORT DATA=work.&File.;
      BY DESCENDING PERCENTSPEND;
   RUN;
%MEND SortTiers;

%SortTiers (File=Tier0);
%SortTiers (File=Tier1);
%SortTiers (File=Tier2);
%SortTiers (File=Tier3);
%SortTiers (File=Tier4);
%SortTiers (File=Tier5);
%SortTiers (File=Tier6);
%SortTiers (File=Tier7);
%SortTiers (File=Tier8);
%SortTiers (File=Tier0Detail);
%SortTiers (File=Tier1Detail);
%SortTiers (File=Tier2Detail);
%SortTiers (File=Tier3Detail);
%SortTiers (File=Tier4Detail);
%SortTiers (File=Tier5Detail);
%SortTiers (File=Tier6Detail);
%SortTiers (File=Tier7Detail);
%SortTiers (File=Tier8Detail);


The next step is to output the 16 data sets to a MS Excel file and once again, for efficiency and coding time savings we build a macro.
*** CREATE MACRO TO EXPORT EACH TIER FILE AND EACH TIER DETAIL FILE TO MS EXCEL ***;

%MACRO ExportProfileData (File=, TabName=   ) ;
   PROC EXPORT DATA=work.&File.
      OUTFILE="I:\SAS\Projects\Provider Profile Report.xlsx"
      LABEL DBMS=XLSX REPLACE;
      SHEET = "&TabName.";
   RUN;
%MEND ExportProfileData;

%ExportProfileData (File=Tier0, TabName=Tier0);
%ExportProfileData (File=Tier1, TabName=Tier1);
%ExportProfileData (File=Tier2, TabName=Tier2);
%ExportProfileData (File=Tier3, TabName=Tier3);
%ExportProfileData (File=Tier4, TabName=Tier4);
%ExportProfileData (File=Tier5, TabName=Tier5);
%ExportProfileData (File=Tier6, TabName=Tier6);
%ExportProfileData (File=Tier7, TabName=Tier7);
%ExportProfileData (File=Tier8, TabName=Tier8);
%ExportProfileData (File=Tier0Detail, TabName=Tier0Detail);
%ExportProfileData (File=Tier1Detail, TabName=Tier1Detail);
%ExportProfileData (File=Tier2Detail, TabName=Tier2Detail);
%ExportProfileData (File=Tier3Detail, TabName=Tier3Detail);
%ExportProfileData (File=Tier4Detail, TabName=Tier4Detail);
%ExportProfileData (File=Tier5Detail, TabName=Tier5Detail);
%ExportProfileData (File=Tier6Detail, TabName=Tier6Detail);
%ExportProfileData (File=Tier7Detail, TabName=Tier7Detail);
%ExportProfileData (File=Tier8Detail, TabName=Tier8Detail);







The last step we output the Prescriber Profile Report which creates 16 tabs, one streamlined with only the metrics per Tier and one with all detail per Tier.  
Below is an example of a TIER 8 prescriber profiling report file, which can be utilized for clinician review.  
	Prescriber ID
	Prescriber Last Name
	Prescriber Formulary Compliance
	Plan Formulary Compliance
	Prescriber Generic Utilization Rate
	Plan Generic Utilization Rate

	99999999999
	DOE
	81.82%
	98.13%
	90.48%
	91.88%



	Prescriber Brand Cost Per Rx
	Plan Brand Cost Per Rx
	Prescriber Generic Cost Per Rx
	Plan Generic Cost Per Rx
	Prescriber Specialty Cost Per Rx
	Plan Specialty Cost Per Rx

	$2,198.26
	$400.01
	$71.29
	$15.04
	$12,830.00
	$5,117.96




	Prescriber Percent Specialty Drugs Prescribed
	Plan Percent Specialty Drugs Prescribed
	Prescriber Percent Hep C Drugs Prescribed
	Plan Percent Hep C Drugs Prescribed
	Prescriber Percent Opioid Drugs Prescribed
	Plan Percent Opioid Drugs Prescribed
	Prescriber Spend Compared to Total Plan

	4.55%
	0.42%
	4.55%
	0.02%
	4.55%
	2.62%
	0.13%



Note in all 8 metrics prescriber Dr. Doe is an outlier and would be a good candidate for review.  The distribution of Prescriber spend in this data set goes up to a max of 1.16%, with 99.24% of prescriber spend at 0.13% or lower, so prescriber spend in this instance is not a red flag.  Therefore more clinical review time might be better spent at lower Tiers with higher prescriber spend percentages.
Next Steps
I have created the same program which compares member metrics to Plan.  Although this is a much larger data set, reviewing the top member outliers using this methodology is an excellent tool, critical to improving patient care and reducing unnecessary cost.

